In[•]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\Theta\\Sage"];
```

pdf

In[•]:=
```
<< KnotTheory`
```

pdf

```
Loading KnotTheory` version of October 29, 2024, 10:29:52.1301.
Read more at http://katlas.org/wiki/KnotTheory.
```

In[•]:=
```
DunfieldKnots =
  ReadList["../../../People/Dunfield/nmd_random_knots"] /. k_Integer :> k + 1;
DK[n_] := DunfieldKnots[[n - 2]];
```

In[•]:=
```
DK[3]
```

Out[•]=
```
PD[X[4, 2, 5, 1], X[6, 4, 1, 3], X[2, 6, 3, 5]]
```

In[•]:=
```
Rot[pd_PD] := Module[{n, xs, x, rots, Xp, Xm, front = {1}, k},
   n = Length@pd; rots = Table[0, {2 n}];
   xs = Cases[pd, x_X :> {Xp[x[[4]], x[[1]]] PositiveQ@x
                          Xm[x[[2]], x[[1]]]      True   ];
   For[k = 1, k ≤ 2 n, ++k,
    If[FreeQ[front, -k],
     front = Flatten@Replace[front, k → (xs /. {
             Xp[k, l_] | Xm[l_, k] :> {l + 1, k + 1, -l},
             Xp[l_, k] | Xm[k, l_] :> (++rots[[l]]; {-l, k + 1, l + 1}),
             _Xp | _Xm :> {}
            }), {1}],
     Cases[front, k | -k] /. {k, -k} :> --rots[[k]];
    ]
   ];
   {xs /. {Xp[i_, j_] :> {+1, i, j}, Xm[i_, j_] :> {-1, i, j}}, rots}];
Rot[K_] := Rot[PD[K]];
```

In[•]:=
```
Rot4Sage[K_] := Module[{Cs, φ},
  {Cs, φ} = Rot[K];
  Cs[[All, 2 ;;]] -= 1;
  AppendTo[φ, 0];
  StringJoin[
   "Cs=matrix(",
   StringReplace[ToString[Cs], {"{" → "[", "}" → "]"}],
   "); phi=",
   StringReplace[ToString[φ], {"{" → "[", "}" → "]"}],
   ";"
  ]
 ]
```

In[•]:=
```
Rot4Sage[DK[15]]
```

Out[•]=
```
Cs=matrix([[-1, 17, 12], [-1, 13, 16], [1, 1, 18], [1, 9, 28], [1, 21, 8], [-1,
  15, 6], [1, 25, 0], [1, 29, 24], [1, 19, 2], [-1, 5, 14], [1, 3, 20], [-1,
  7, 4], [1, 26, 11], [1, 22, 27], [1, 10, 23]]); phi=[0, 0, 0, 0, 0, 0, 0,
  -1, 0, 0, 0, 0, 0, 0, 0, 0, 1, -1, -1, 0, 0, 0, 0, 0, 0, 1, 1, -1, 0, 0, 0];
```

In[•]:=

```
PolyPlot1[Δ_] := Module[{crs, m, maxc, minc, s, rect},
   rect = {{0, 0}, {1, 0}, {1, 1}, {0, 1}};
   If[Expand[Δ] === 0, Graphics[],
    m = Max[-Exponent[Δ, T, Min], Exponent[Δ, T, Max]];
    crs = CoefficientRules[T^m Δ, {T}];
    maxc = N@Log@Max@Abs[Last /@ crs];
    minc = N@Log@Min@Select[Abs[Last /@ crs], # > 0 &];
    If[minc == maxc, s[_] = 0, s[c_] := s[c] = (maxc - Log@c) / (maxc - minc)];
    Graphics[crs /. ({x_} → c_) :> {
        Lighter[Which[c == 0, White, c > 0, Red, c < 0, Blue], 0.88 s[Abs@c]],
        Tooltip[Polygon[({x + m - 1 / 2, 0} + #) & /@ rect], c T^(x-m)]
       }, AspectRatio → Min[1 / 5, 1 / Sqrt[m + 1]],
     ImagePadding → None, PlotRangePadding → None]
   ]];
Options[PolyPlot2] = {Labeled → False};
PolyPlot2[Θ_, OptionsPattern[]] := Module[{crs, m1, m2, maxc, minc, s, hex, p},
   If[Expand[Θ] === 0, Graphics[{White, Disk[]}],
    hex = Table[{Cos[α], Sin[α]} / Cos[2 π / 12] / 2, {α, 2 π / 12, 2 π, 2 π / 6}];
    m1 = Max[-Exponent[Θ, T1, Min], Exponent[Θ, T1, Max]];
    m2 = Max[-Exponent[Θ, T2, Min], Exponent[Θ, T2, Max]];
    crs = CoefficientRules[T1^m1 T2^m2 Θ, {T1, T2}];
    maxc = N@Log@Max@Abs[Last /@ crs];
    minc = N@Log@Min@Select[Abs[Last /@ crs], # > 0 &];
    If[minc == maxc, s[_] = 0, s[c_] := s[c] = (maxc - Log@c) / (maxc - minc)];
    Graphics[{(*{Yellow,Disk[{0,0},1+Cos[2π/12]Norm[{m1,m2}]/Sqrt[2]]},*)
      crs /. ({x1_, x2_} → c_) :> {
         Lighter[Which[c == 0, White, c > 0, Red, c < 0, Blue], 0.88 s[Abs@c]],
         p = {{1, -1 / 2}, {0, Sqrt[3] / 2}}.{x1 - m1, x2 - m2};
         Tooltip[Polygon[(p + #) & /@ hex], c T1^(x1-m1) T2^(x2-m2)],
         If[Not@OptionValue[Labeled], {}, {Black, Text[Style[c T1^(x1-m1) T2^(x2-m2), 30], p]}]
        }
     }, ImagePadding → None, PlotRangePadding → None]
   ]];
PolyPlot[{Δ_, Θ_}, opts___Rule] := GraphicsColumn[
   {PolyPlot1[Δ], PolyPlot2[Θ, FilterRules[{opts}, Options[PolyPlot2]]]},
   Spacings → Scaled@0.08, ImagePadding → None, PlotRangePadding → None,
   FilterRules[{opts}, Options[GraphicsColumn]]
  ];
```